Introduction to Maple and Linear Algebra

Mth 341 - Summer 1995

Oct 26 1996

These notes are a compilation of seven sets of notes distributed in Mth 341, Linear Algebra, during the OSU 1995 Summer Session. The original notes were created in Maple V3, exported as LaTeX, edited and converted to $\mathcal{A}\mathcal{M}\mathcal{S}$-LaTeX. This compilation was lightly edited, converted to LaTeX 2_ε and compiled with the Maple V4 LaTeX style file.

It should be realized that some of the more idiosyncratic behavior of Maple V3 referred to below may have changed in Maple V4, or may change in future versions.

1 Handout a

Before doing any linear algebra with maple it's a good idea to load the *linalg* package:

```
> with(linalg):

Warning: new definition for   norm
Warning: new definition for   trace
```

The colon which terminates this command supresses output. Without it we'd get a blitz of messages. The redefinition error messages can be ignored.

Matrices can be defined as follows:

```
> A := matrix(3,6,[0,3,-6,6,4,-5,3,-7,8,-5,8,9,3,-9,12,-9,6,15]);
```

$$A := \begin{bmatrix} 0 & 3 & -6 & 6 & 4 & -5 \\ 3 & -7 & 8 & -5 & 8 & 9 \\ 3 & -9 & 12 & -9 & 6 & 15 \end{bmatrix}$$

The first two parameters specify the size of the matrix. The third parameter is a list (indicated by square brackets) of the entries in the matrix, in row-major order (that is, row-by-row).

Maple has a built-in command to compute the reduced row echelon canonical form:

```
> rref(A);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & 0 & -24 \\ 0 & 1 & -2 & 2 & 0 & -7 \\ 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix}$$

A synonym for rref() (reduced row echelon form) is gaussjord() (Gauss-Jordan form):

```
> gaussjord(A);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & 0 & -24 \\ 0 & 1 & -2 & 2 & 0 & -7 \\ 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix}$$

We can also do the row reduction *by hand*:

```
> swaprow(A,1,2);
```

$$\begin{bmatrix} 3 & -7 & 8 & -5 & 8 & 9 \\ 0 & 3 & -6 & 6 & 4 & -5 \\ 3 & -9 & 12 & -9 & 6 & 15 \end{bmatrix}$$

```
> mulrow(",1,1/3);
```

$$\begin{bmatrix} 1 & \frac{-7}{3} & \frac{8}{3} & \frac{-5}{3} & \frac{8}{3} & 3 \\ 0 & 3 & -6 & 6 & 4 & -5 \\ 3 & -9 & 12 & -9 & 6 & 15 \end{bmatrix}$$

Here the double quote is the ditto operator – Maple interprets it to mean the previous expression: in our case the matrix A.

```
> addrow(",1,3,-3);
```

$$\begin{bmatrix} 1 & \frac{-7}{3} & \frac{8}{3} & \frac{-5}{3} & \frac{8}{3} & 3 \\ 0 & 3 & -6 & 6 & 4 & -5 \\ 0 & -2 & 4 & -4 & -2 & 6 \end{bmatrix}$$

```
> addrow(",3,2,1);
```

$$\begin{bmatrix} 1 & \frac{-7}{3} & \frac{8}{3} & \frac{-5}{3} & \frac{8}{3} & 3 \\ 0 & 1 & -2 & 2 & 2 & 1 \\ 0 & -2 & 4 & -4 & -2 & 6 \end{bmatrix}$$

```
> addrow(",2,1,7/3);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & \frac{22}{3} & \frac{16}{3} \\ 0 & 1 & -2 & 2 & 2 & 1 \\ 0 & -2 & 4 & -4 & -2 & 6 \end{bmatrix}$$

```
> addrow(",2,3,2);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & \frac{22}{3} & \frac{16}{3} \\ 0 & 1 & -2 & 2 & 2 & 1 \\ 0 & 0 & 0 & 0 & 2 & 8 \end{bmatrix}$$

```
> mulrow(",3,1/2);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & \frac{22}{3} & \frac{16}{3} \\ 0 & 1 & -2 & 2 & 2 & 1 \\ 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix}$$

```
> addrow(",3,1,-22/3);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & 0 & -24 \\ 0 & 1 & -2 & 2 & 2 & 1 \\ 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix}$$

```
> addrow(",3,2,-2);
```

$$\begin{bmatrix} 1 & 0 & -2 & 3 & 0 & -24 \\ 0 & 1 & -2 & 2 & 0 & -7 \\ 0 & 0 & 0 & 0 & 1 & 4 \end{bmatrix}$$

Sure beats doing the arithmetic by hand!

2 Handout b

Here's some more Maple examples.

```
> with(linalg):

Warning: new definition for   norm
Warning: new definition for   trace

> A:=matrix([[0,-3,-6,4],[-1,-2,-1,3],[-2,-3,0,3],[1,4,5,-9]]);
```

$$A := \begin{bmatrix} 0 & -3 & -6 & 4 \\ -1 & -2 & -1 & 3 \\ -2 & -3 & 0 & 3 \\ 1 & 4 & 5 & -9 \end{bmatrix}$$

Note here we defined the matrix A by feeding a list of its rows to the matrix() function. We didn't have to specify the size of the matrix. Sometimes though it's more convenient to specify the size, as for example:

```
> b:=matrix(4,1,[9,1,-1,-7]);
```

$$b := \begin{bmatrix} 9 \\ 1 \\ -1 \\ -7 \end{bmatrix}$$

To solve the system of equations $Ax = b$ we can form the augmented matrix

```
> C:=augment(A,b);
```

$$C := \begin{bmatrix} 0 & -3 & -6 & 4 & 9 \\ -1 & -2 & -1 & 3 & 1 \\ -2 & -3 & 0 & 3 & -1 \\ 1 & 4 & 5 & -9 & -7 \end{bmatrix}$$

and row reduce

```
> rref(C);
```

$$\begin{bmatrix} 1 & 0 & -3 & 0 & 5 \\ 0 & 1 & 2 & 0 & -3 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

At this point we can read-off the solutions. As you might expect though, Maple does provide a facility for solving linear equations directly:

```
> linsolve(A,b);
```

$$\begin{bmatrix} 5+3\_t_1 \\ -3-2\_t_1 \\ \_t_1 \\ 0 \end{bmatrix}$$

Here $\_t_1$ is a parameter; let's denote it by t since the underscore is unaesthetic and we don't really need the subscript. Then Maple's response means

$$\begin{aligned} x_1 &= 3t+5 \\ x_2 &= -2t-3 \\ x_3 &= t \\ x_4 &= 0 \end{aligned}$$

Let's see what happens when there's no solution:

```
> A:=matrix(2,2,[1,1,1,1]); b:=matrix(2,1,[1,2]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}$$

$$b := \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

```
> linsolve(A,b);
```

Maple appears to be silent, but has in fact returned the empty set!

We can generate random matrices (useful for examples):

```
> A:=randmatrix(4,5); b:=randmatrix(4,1);
```

$$A := \begin{bmatrix} -85 & -55 & -37 & -35 & 97 \\ 50 & 79 & 56 & 49 & 63 \\ 57 & -59 & 45 & -8 & -93 \\ 92 & 43 & -62 & 77 & 66 \end{bmatrix}$$

$$b := \begin{bmatrix} 54 \\ -5 \\ 99 \\ -61 \end{bmatrix}$$

```
> linsolve(A,b);
```

$$\begin{bmatrix} -\frac{13620907}{13630726} + \frac{78454729}{27261452}\_t_1 \\ -\frac{13466201}{6815363} + \frac{19319399}{13630726}\_t_1 \\ \frac{18178923}{13630726} - \frac{19875703}{27261452}\_t_1 \\ \frac{17576855}{6815363} - \frac{38671600}{6815363}\_t_1 \\ \_t_1 \end{bmatrix}$$

The fact that we have one parameter here tells us that there is only one free variable. The remaining 4 variables are bound (pivotal). Thus the rank of A must be 4. Let's see if Maple agrees:

```
> rank(A);
```

$$4$$

Sure enough, we were right about the rank. Note that Maple gave us the exact solution above. Sometimes this is inconvenient. We can round off the answer and express it in decimal notation as follows:

```
> evalf("");
```

$$\begin{bmatrix} -.9992796422 + 2.877863182\_t_1 \\ -1.975859686 + 1.417341894\_t_1 \\ 1.333672396 - .7290771966\_t_1 \\ 2.579004963 - 5.674180524\_t_1 \\ \_t_1 \end{bmatrix}$$

The double quote here refers to the expression prior to the previous one.

3 Handout c

In this note we will see (by example) how to determine if a given vector is a linear combination of other given vectors.

```
> with(linalg):
```

```
Warning: new definition for   norm
Warning: new definition for   trace
```

Consider the following vectors

```
> u:=matrix(4,1,[2,1,-3,1]):  v:=matrix(4,1,[-1,3,5,0]):  w:=matrix(4,1,[2,-1,1,-3]):
> b:=matrix(4,1,[-16,17,37,3]);
```

$$b := \begin{bmatrix} -16 \\ 17 \\ 37 \\ 3 \end{bmatrix}$$

We'd like to know if we can write b as a linear combination of u, v and w, that is can we find scalars x_1, x_2, x_3 such that $b = x_1\, u + x_2\, v + x_3\, w$. If we let A be the matrix with columns u, v, w

```
> A := augment(u,v,w);
```

$$A := \begin{bmatrix} 2 & -1 & 2 \\ 1 & 3 & -1 \\ -3 & 5 & 1 \\ 1 & 0 & -3 \end{bmatrix}$$

then we see that we are looking for a solution to the system $Ax = b$.

```
> linsolve(A,b);
```

$$\begin{bmatrix} -3 \\ 6 \\ -2 \end{bmatrix}$$

Thus $b = -3\, u + 6\, v - 2\, w$.

4 Handout d

In this handout we look at the problem of finding a one-sided inverse. First let's look at the theory:

Theorem 1. *Let A be an $m \times n$ matrix. Then A has a left inverse if and only if the $n \times n$ matrix $A^T A$ is invertible. If A has a left inverse then $m \geq n$ and $\left(A^T A\right)^{-1} A^T$ is a left inverse for A.*

Proof. If $A^T A$ is invertible then $\left(A^T A\right)^{-1} A^T A = I$ implies $D = \left(A^T A\right)^{-1} A^T$ is a left inverse for A. Conversely suppose D is any left inverse for A (in general not unique). Let $\vec{x} \in \mathbb{R}^n$ and suppose $A^T A\vec{x} = \vec{0}$. Then $\vec{0} = \vec{x}^T A^T A\vec{x} = (A\vec{x})^T (A\vec{x})$ implies $A\vec{x} = \vec{0}$. But then $\vec{x} = DA\vec{x} = D\vec{0} = \vec{0}$. Thus $A^T A\vec{x} = \vec{0}$ implies $\vec{x} = \vec{0}$ and therefore the square matrix $A^T A$ is invertible. If A has a left inverse D then $DA\vec{x} = \vec{x}$ implies $A\vec{x} = \vec{0}$ has only the trivial solution. Hence each column in A must be pivotal and so $m \geq n$. □

Theorem 2. *Let A be an $m \times n$ matrix. Then A has a right inverse if and only if the $m \times m$ matrix AA^T is invertible. If A has a right inverse then $n \geq m$ and $A^T \left(AA^T\right)^{-1}$ is a right inverse for A.*

Proof. If D is a right inverse for A then D^T is a left inverse for A^T. □

We can use the explicit formulae above to find left or right inverses. Keep in mind thought that these are usually not unique – we are selecting a particular one.

```
> with(linalg):
```

```
Warning: new definition for   norm
Warning: new definition for   trace
```

```
> A:=matrix(2,3,[1,2,-1,3,1,2]);
```

$$A := \begin{bmatrix} 1 & 2 & -1 \\ 3 & 1 & 2 \end{bmatrix}$$

Since A is 2×3 we have $m < n$ and so A cannot have a left inverse. To check for a right inverse we compute AA^T:

```
> B:=evalm(A &* transpose(A));
```

$$B := \begin{bmatrix} 6 & 3 \\ 3 & 14 \end{bmatrix}$$

```
> rref(B);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Since the reduced row echelon form of B is I we know B is invertible. Thus A has a right inverse given by $C = A^T B^{-1}$:

```
> C:=evalm(transpose(A) &* inverse(B));
```

$$C := \begin{bmatrix} \frac{1}{15} & \frac{1}{5} \\ \frac{1}{3} & 0 \\ \frac{-4}{15} & \frac{1}{5} \end{bmatrix}$$

We can check our proposed right inverse by multiplying:

```
>  evalm(A &* C);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

It works!

Note Maple uses &* for matrix multiplication, not *. The reason is that matrix multiplication is not commutative. Adding the ampersand prevents Maple from attempting simplifications which assume commutativity.

Exercise 1. *We commented above that A cannot have a left inverse. It follows that the 3×3 matrix $A^T A$ is not invertible. Verify this fact by computing $A^T A$ and row-reducing it. (Use Maple, of course.)*

5 Handout e

Let's look again at the problem of finding right inverses. (We can handle left inverses by considering the transpose.)

```
>  with(linalg):
Warning: new definition for   norm
Warning: new definition for   trace
```

Consider the same example as in handout d:

```
>  A:=matrix(2,3,[1,2,-1,3,1,2]);
```

$$A := \begin{bmatrix} 1 & 2 & -1 \\ 3 & 1 & 2 \end{bmatrix}$$

```
>  Id:=diag(1,1);
```

$$Id := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

To find a right inverse for A we solve the equation $AB = I$. Maple can solve this equation with the same function call as is used to solve $A\vec{x} = \vec{b}$.

```
>  B:=linsolve(A,Id);
```

$$B := \begin{bmatrix} -\frac{1}{5} - \_t_1 & \frac{2}{5} - \_t_1 \\ \frac{3}{5} + \_t_1 & -\frac{1}{5} + \_t_1 \\ \_t_1 & \_t_1 \end{bmatrix}$$

Well this is certainly correct - it is a solution - but Maple doesn't seem to realize that the two columns can be selected independently of each other. The solution really should contain 2 parameters. We use the subs() function (substitution) to replace Maple's one parameter with 2 parameters, s in the first column, and t in the second column.

```
> C:=(s,t)->augment(subs(_t[1]=s,col(B,1)), subs(_t[1]=t,col(B,2)));
```

$$C := (s,t) \to \operatorname{augment}(\operatorname{subs}(\_t_1 = s, \operatorname{col}(B,1)), \operatorname{subs}(\_t_1 = t, \operatorname{col}(B,2)))$$

What a mess! Does it really look like it's supposed to do?

```
> C(s,t);
```

$$\begin{bmatrix} -\frac{1}{5} - s & \frac{2}{5} - t \\ \frac{3}{5} + s & -\frac{1}{5} + t \\ s & t \end{bmatrix}$$

Let's check that it works:

```
> evalm( A &* C(s,t));
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Sure enough $C(s,t)$ is a right inverse for any s and any t.

Taking $s = -4/15$ and $t = 1/5$ we get the same solution as in handout d.

```
> C(-4/15,1/5);
```

$$\begin{bmatrix} \frac{1}{15} & \frac{1}{5} \\ \frac{1}{3} & 0 \\ \frac{-4}{15} & \frac{1}{5} \end{bmatrix}$$

We could never have obtained this solution directly from Maple's solution by substituting for $\_t_1$. Even though Maple is quite clever we do have to coax it a bit sometimes.

6 Handout f

In this handout we simply illustrate Maple's functions for computing eigenvalues and eigenvectors.

```
> with(linalg):

Warning: new definition for   norm
Warning: new definition for   trace
```

```
> A:=matrix(3,3,[-1,4,-2,-3,4,0,-3,1,3]);
```

$$A := \begin{bmatrix} -1 & 4 & -2 \\ -3 & 4 & 0 \\ -3 & 1 & 3 \end{bmatrix}$$

```
> eigenvals(A);
```

$$1, 2, 3$$

```
> eigenvects(A);
```

$$[1, 1, \{[1\,1\,1]\}], \left[2, 1, \left\{\left[1\ \frac{3}{2}\ \frac{3}{2}\right]\right\}\right], [3, 1, \{[1\,3\,4]\}]$$

The format here is [eigenvalue, algebraic multiplicity, {set of linearly independent eigenvectors}].

```
> P:=transpose(matrix(3,3,[1,1,1,1,3/2,3/2,1,3,4]));
```

$$P := \begin{bmatrix} 1 & 1 & 1 \\ 1 & \frac{3}{2} & 3 \\ 1 & \frac{3}{2} & 4 \end{bmatrix}$$

```
> evalm(inverse(P) &* A &* P);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> B:=matrix(3,3,[4,2,2,2,4,2,2,2,4]);
```

$$B := \begin{bmatrix} 4 & 2 & 2 \\ 2 & 4 & 2 \\ 2 & 2 & 4 \end{bmatrix}$$

```
> eigenvals(B);
```

$$8, 2, 2$$

```
> eigenvects(B);
```

$$[8, 1, \{[1\,1\,1]\}], [2, 2, \{[-1\,1\,0], [-1\,0\,1]\}]$$

Note we have 2 linearly independent eigenvectors for the eigenvalue 2 and so 3 linearly independent eigenvectors all together. Thus B is diagonizable.

```
> Q:=transpose(matrix(3,3,[1,1,1,-1,1,0,-1,0,1]));
```

$$Q := \begin{bmatrix} 1 & -1 & -1 \\ 1 & 1 & 0 \\ 1 & 0 & 1 \end{bmatrix}$$

```
> evalm(inverse(Q) &* B &* Q);
```

$$\begin{bmatrix} 8 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

We can also use Maple to compute characteristic polynomials:

```
> charpoly(A,lambda);
```

$$\lambda^3 - 6\,\lambda^2 + 11\,\lambda - 6$$

```
> charpoly(B,lambda);
```

$$\lambda^3 - 12\,\lambda^2 + 36\,\lambda - 32$$

Obviously Maple uses $\det(\lambda I - A)$ for the chararcteristic polynomial rather than $\det(A - \lambda I)$ as we did in class.

The question of diagonizability can be answered by computing the Jordan normal form - this is a matrix similar to the given one and as close to being diagonal as possible. Thus A is diagonizable if and only if jordan(A) is diagonal. In any case, the diagonal entries in jordan(A) are just the eigenvalues of A, repeated according to algebraic multiplicity.

```
> jordan(A);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> jordan(B);
```

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & 8 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

```
> C:=matrix(3,3,[1,2,1,3,3,3,2,1,2]);
```

$$C := \begin{bmatrix} 1 & 2 & 1 \\ 3 & 3 & 3 \\ 2 & 1 & 2 \end{bmatrix}$$

```
> jordan(C);
```

$$\begin{bmatrix} 6 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}$$

We see the eigenvalues of C are 6,0,0 and C is not diagonizable. Let's compute the eigenvectors of C:

```
> eigenvects(C);
```

$$[6, 1, \{[1\,2\,1]\}], [0, 2, \{[-1\,0\,1]\}]$$

Sure enough - we have only 2 linearly independent eigenvectors. Thus C is not diagonizable.

7 Handout g

In this handout we continue to illustrate Maple's facilities for computing eigenvalues and eigenvectors.

```
> with(linalg):

Warning: new definition for   norm
Warning: new definition for   trace

> A:=matrix(3,3,[1,2,4,3,7,2,5,6,9]);
```

$$A := \begin{bmatrix} 1 & 2 & 4 \\ 3 & 7 & 2 \\ 5 & 6 & 9 \end{bmatrix}$$

```
> ev:=eigenvals(A);
```

$$ev := \%1^{1/3} + \frac{166}{9}\frac{1}{\%1^{1/3}} + \frac{17}{3},$$
$$-\frac{1}{2}\%1^{1/3} - \frac{83}{9}\frac{1}{\%1^{1/3}} + \frac{17}{3} + \frac{1}{2}I\sqrt{3}\left(\%1^{1/3} - \frac{166}{9}\frac{1}{\%1^{1/3}}\right),$$
$$-\frac{1}{2}\%1^{1/3} - \frac{83}{9}\frac{1}{\%1^{1/3}} + \frac{17}{3} - \frac{1}{2}I\sqrt{3}\left(\%1^{1/3} - \frac{166}{9}\frac{1}{\%1^{1/3}}\right)$$
$$\%1 := \frac{1088}{27} + \frac{2}{9}I\sqrt{94182}$$

Here's one place where Maple's penchant for *exact* answers get's in the way. We can convert the exact answer to an approximate floating point answer:

```
> evalf(ev);
```

$$13.74788906, -.894602542 + .12\,10^{-8}\,I, 4.146713484 - .6\,10^{-9}\,I$$

The symbol I is the square root of -1. Thus we have complex answers with very small imaginary parts. Let's increase the accuracy:

```
> Digits:=18:
> evalf(ev);
```

$$13.7478890587268580, -.89460254283571530 - .1\,10^{-17}\,I,$$
$$4.14671348410885728 - .1\,10^{-17}\,I$$

The imaginary parts are smaller! In fact they are of the same order as the precision of our conversion. This is pretty sure sign we are seeing a round-off error. If we really want a floating point answer we can force Maple to provide one from the beginning by putting at least one floating point number in the matrix:

```
> Digits:=10:
> B:=matrix(3,3,[1.0,2,4,3,7,2,5,6,9]);
```

$$B := \begin{bmatrix} 1.0 & 2 & 4 \\ 3 & 7 & 2 \\ 5 & 6 & 9 \end{bmatrix}$$

```
> eigenvals(B);
```

$$-.8946025434, 13.74788901, 4.146713483$$

This answer less precise than the one we obtained above (with the default precision of 10 digits) however we no longer have the troublesome imaginary parts.

What about eigenvectors?

```
> evt:=eigenvects(A);
```

$$evt := \left[\%1, 1, \left\{\left[\frac{53}{61}\,\%1 - \frac{138}{61} - \frac{3}{61}\,\%1^2 \;\; -\frac{34}{61}\,\%1 + \frac{47}{122} + \frac{5}{122}\,\%1^2 \;\; 1\right]\right\}\right]$$
$$\%1 := \mathrm{RootOf}(\_Z^3 - 17\,\_Z^2 + 41\,\_Z + 51)$$

Here %1 runs over the roots of the characteristic polynomial, that is, the eigenvalues. We can try a floating point conversion again:

```
> evalf(evt);
```

$$[-.8946025428, 1., \{[-3.078932392\ .9166765693\ 1.]\}]$$

Sigh. We only get one of the roots and the corresponding eigenvector. More work is needed. On the other hand if we force Maple to work in floating point from the beginning we get all the eigenvectors:

```
> eigenvects(B);
```

$$[\,13.74788901, 1, \{\,[\,.3592673648\ .4346670837\ .9276415841\,]\,\}\,],$$
$$[\,4.146713483, 1, \{\,[\,-.2987282688\ .7371720079\ -.6035890728\,]\,\}\,],$$
$$[\,-.8946025434, 1, \{\,[\,.946575123\ -.2818195168\ -.3074361518\,]\,\}\,]$$

Actually if we want to compute eigenvalues and eigenvectors in floating point the best way is to use the inert (unevaluated) function Eigenvals() - note the capitol 'E'. This function is designed for numeric matrices. We force evaluation by using the evalf() function:

```
> evalf(Eigenvals(B));
```

$$[\,-.8946025434\ 13.74788901\ 4.146713483\,]$$

We can also get the eigenvectors simply by providing a second variable:

```
> evalf(Eigenvals(B,C));
```

$$[\,-.8946025434\ 13.74788901\ 4.146713483\,]$$

```
> R:=evalm(C);
```

$$R := \begin{bmatrix} .946575123 & .3592673648 & -.2987282688 \\ -.2818195168 & .4346670837 & .7371720079 \\ -.3074361518 & .9276415841 & -.6035890728 \end{bmatrix}$$

If we compute $R^{-1}BR$ we should get a diagonal matrix with the eigenvalues of B on the diagonal:

```
> evalm( inverse(R) &* B &* R);
```

$$\begin{bmatrix} -.8946025428 & -.45\,10^{-8} & -.13\,10^{-8} \\ -.17\,10^{-7} & 13.74788906 & -.7\,10^{-8} \\ -.21\,10^{-8} & .6\,10^{-8} & 4.146713484 \end{bmatrix}$$

Apart from a fairly large round-off error we get what we expected.

Bent E. Petersen
Department of Mathematics
Oregon State University
Corvallis, OR 97331-4605

24 hour phone numbers
office (541) 737-5163
home (541) 754-2315
fax (541) 737-0517

email: petersen@math.orst.edu
web: http://www.orst.edu/~peterseb